<commit_before>plugins {
    `ivy-publish`
}

group = "org.gradle.sample"
version = "1.0"

publishing {
    // tag::customize-descriptor[]
    publications {
        create<IvyPublication>("ivyCustom") {
            descriptor {
                license {
                    name.set("The Apache License, Version 2.0")
                    url.set("http://www.apache.org/licenses/LICENSE-2.0.txt")
                }
                author {
                    name.set("<NAME>")
                    name.set("http://example.com/users/jane")
                }
                description {
                    text.set("A concise description of my library")
                    homepage.set("http://www.example.com/library")
                }
            }
        }
    }
// end::customize-descriptor[]
    repositories {
        ivy {
            // change to point to your repo, e.g. http://my.org/repo
            url = uri("$buildDir/repo")
        }
    }
}

// tag::generate[]
tasks.getByName<GenerateIvyDescriptor>("generateDescriptorFileForIvyCustomPublication") {
    destination = file("$buildDir/generated-ivy.xml")
}
// end::generate[]<commit_msg>Fix ivy-publish/descriptor-customization/kotlin sample<commit_after>plugins {
    `ivy-publish`
}

group = "org.gradle.sample"
version = "1.0"

publishing {
    // tag::customize-descriptor[]
    publications {
        create<IvyPublication>("ivyCustom") {
            descriptor {
                license {
                    name.set("The Apache License, Version 2.0")
                    url.set("http://www.apache.org/licenses/LICENSE-2.0.txt")
                }
                author {
                    name.set("<NAME>")
                    url.set("http://example.com/users/jane")
                }
                description {
                    text.set("A concise description of my library")
                    homepage.set("http://www.example.com/library")
                }
            }
        }
    }
// end::customize-descriptor[]
    repositories {
        ivy {
            // change to point to your repo, e.g. http://my.org/repo
            url = uri("$buildDir/repo")
        }
    }
}

// tag::generate[]
tasks.getByName<GenerateIvyDescriptor>("generateDescriptorFileForIvyCustomPublication") {
    destination = file("$buildDir/generated-ivy.xml")
}
// end::generate[]
{'fraction_non_alphanumeric': 0.13570822731128074, 'fraction_numerical': 0.005089058524173028, 'mean_word_length': 1.5613463626492943, 'pattern_counts': {'":': 0, '<': 9, '<?xml version=': 0, '>': 9, 'https://': 0, 'lorem ipsum': 0, 'www.': 4, 'xml': 2}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4049314', 'n_tokens_mistral': 675, 'n_tokens_neox': 651, 'n_words': 95}